# $Id: PKGBUILD 88702 2013-04-21 22:24:11Z heftig $
# Maintainer: Roman Kyrylych <roman@archlinux.org>

pkgname=gnome-phone-manager
pkgver=0.69
pkgrel=1
pkgdesc="Control your mobile phone from your GNOME desktop."
arch=('i686' 'x86_64')
url="http://live.gnome.org/PhoneManager"
license=('GPL')
depends=('evolution-data-server' 'gnokii' 'gnome-bluetooth' 'gnome-icon-theme' 'libcanberra' 'gconf')
makedepends=('intltool')
options=('!emptydirs' '!libtool')
install=gnome-phone-manager.install
source=(http://ftp.gnome.org/pub/GNOME/sources/$pkgname/$pkgver/$pkgname-$pkgver.tar.xz
        gnome-phone-manager-0.68-eds.patch)
sha256sums=('35e038ea3afaacdf451046e87af876096cf1520efc04fc3f5b63ea22e0297175'
            '232a72108836ddd46d329993d166c38550e2923f161b4dc814d75af4ed5ccc79')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -p1 -i ../gnome-phone-manager-0.68-eds.patch
  ./configure --prefix=/usr --sysconfdir=/etc \
    --localstatedir=/var --disable-static
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" GCONF_DISABLE_MAKEFILE_SCHEMA_INSTALL=1 install

  install -m755 -d "${pkgdir}/usr/share/gconf/schemas"
  gconf-merge-schema "${pkgdir}/usr/share/gconf/schemas/${pkgname}.schemas" --domain gnome-phone-manager "${pkgdir}"/etc/gconf/schemas/*.schemas
  rm -f "${pkgdir}"/etc/gconf/schemas/*.schemas
}
